Die freie Schwarzschildlösung

Allgemeinster kugelsymmetrischer Ansatz für die Metrik

```
q = {t, r, th, ph};
```

```
g = {{l[r, t], a[r, t] / 2, 0, 0},
   {a[r, t] / 2, h[r, t], 0, 0}, {0, 0, k[r, t], 0}, {0, 0, 0, k[r, t] Sin[th] ^2}};
```

```
MatrixForm[g]
```

$$\begin{pmatrix} l[r,t] & \frac{1}{2}a[r,t] & 0 & 0 \\ \frac{1}{2}a[r,t] & h[r,t] & 0 & 0 \\ 0 & 0 & k[r,t] & 0 \\ 0 & 0 & 0 & k[r,t]\,\mathrm{Sin}[th]^2 \end{pmatrix}$$

```
dq = {dt, dr, dth, dph};
```

```
FullSimplify[dq.g.dq]
```

$$dr\ dt\ a[r,t] + dr^2\ h[r,t] + dt^2\ l[r,t] + k[r,t]\left(dth^2 + dph^2\ \mathrm{Sin}[th]^2\right)$$

Durch geeignete Umparametrisierung von r und t können wir die Außerdiagonalelemente zum Verschwinden bringen und k=-r^2 setzen:

```
h[r_, t_] = -Exp[la[r, t]]; l[r_, t_] = c ^2 Exp[nu[r, t]]; a[r_, t_] = 0; k[r_, t_] = -r ^2;
```

```
FullSimplify[dq.g.dq]
```

$$-dr^2\ e^{la[r,t]} + c^2\ dt^2\ e^{nu[r,t]} - dth^2\ r^2 - dph^2\ r^2\ \mathrm{Sin}[th]^2$$

```
gcontra = Inverse[g];
```

Wir lassen und als nächstes die Christoffelsymbole für den Riemannschen affinen Zusammenhang aus der Definition durch Ableitungen der Metrik ausrechnen:

```
christ = Table[Table[Sum[1 / 2 gcontra[[ii]][[mi]]
      (D[g[[mi]][[ki]], q[[li]]] + D[g[[mi]][[li]], q[[ki]]] - D[g[[ki]][[li]], q[[mi]]]),
     {mi, 1, 4}], {ki, 1, 4}, {li, 1, 4}], {ii, 1, 4}];
```

```
Do[Print[MatrixForm[christ[[ii]]]], {ii, 1, 4}]
```

$$\begin{pmatrix} \frac{1}{2}\,\text{nu}^{(0,1)}[r,t] & \frac{1}{2}\,\text{nu}^{(1,0)}[r,t] & 0 & 0 \\ \frac{1}{2}\,\text{nu}^{(1,0)}[r,t] & \frac{e^{\text{la}[r,t]-\text{nu}[r,t]}\,\text{la}^{(0,1)}[r,t]}{2\,c^2} & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

$$\begin{pmatrix} \frac{1}{2}\,c^2\,e^{-\text{la}[r,t]+\text{nu}[r,t]}\,\text{nu}^{(1,0)}[r,t] & \frac{1}{2}\,\text{la}^{(0,1)}[r,t] & 0 & 0 \\ \frac{1}{2}\,\text{la}^{(0,1)}[r,t] & \frac{1}{2}\,\text{la}^{(1,0)}[r,t] & 0 & 0 \\ 0 & 0 & -e^{-\text{la}[r,t]}\,r & 0 \\ 0 & 0 & 0 & -e^{-\text{la}[r,t]}\,r\,\text{Sin}[\text{th}]^2 \end{pmatrix}$$

$$\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{1}{r} & 0 \\ 0 & \frac{1}{r} & 0 & 0 \\ 0 & 0 & 0 & -\text{Cos}[\text{th}]\,\text{Sin}[\text{th}] \end{pmatrix}$$

$$\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{1}{r} \\ 0 & 0 & 0 & \text{Cot}[\text{th}] \\ 0 & \frac{1}{r} & \text{Cot}[\text{th}] & 0 \end{pmatrix}$$

Daraus ergibt sich sofort für den Ricci-Tensor (vollständig kovariante Komponenten)

```
RicciTen = FullSimplify[
  Table[Sum[D[christ[[li]][[ii]][[ki]], q[[li]]] - D[christ[[li]][[ii]][[li]], q[[ki]]] +
     Sum[christ[[li]][[ii]][[ki]] christ[[mi]][[li]][[mi]] - christ[[mi]][[ii]][[li]]
        christ[[li]][[ki]][[mi]], {mi, 1, 4}], {li, 1, 4}], {ii, 1, 4}, {ki, 1, 4}]]
```

$$\Big\{\Big\{\frac{1}{4\,r}\,e^{-\text{la}[r,t]}\,\Big(-e^{\text{la}[r,t]}\,r\,\big(\text{la}^{(0,1)}[r,t]^2-\text{la}^{(0,1)}[r,t]\,\text{nu}^{(0,1)}[r,t]+2\,\text{la}^{(0,2)}[r,t]\big)+$$
$$c^2\,e^{\text{nu}[r,t]}\,\big(\text{nu}^{(1,0)}[r,t]\,\big(4-r\,\text{la}^{(1,0)}[r,t]+r\,\text{nu}^{(1,0)}[r,t]\big)+2\,r\,\text{nu}^{(2,0)}[r,t]\big)\Big),$$
$$\frac{\text{la}^{(0,1)}[r,t]}{r},\,0,\,0\Big\},\,\Big\{\frac{\text{la}^{(0,1)}[r,t]}{r},$$
$$\frac{1}{4}\,\Big(\frac{1}{c^2}\,e^{\text{la}[r,t]-\text{nu}[r,t]}\,\big(\text{la}^{(0,1)}[r,t]^2-\text{la}^{(0,1)}[r,t]\,\text{nu}^{(0,1)}[r,t]+2\,\text{la}^{(0,2)}[r,t]\big)+$$
$$\frac{1}{r}\,\big(\text{la}^{(1,0)}[r,t]\,\big(4+r\,\text{nu}^{(1,0)}[r,t]\big)-r\,\big(\text{nu}^{(1,0)}[r,t]^2+2\,\text{nu}^{(2,0)}[r,t]\big)\big)\Big),\,0,\,0\Big\},$$
$$\Big\{0,\,0,\,1+\frac{1}{2}\,e^{-\text{la}[r,t]}\,\big(-2+r\,\text{la}^{(1,0)}[r,t]-r\,\text{nu}^{(1,0)}[r,t]\big),\,0\Big\},$$
$$\Big\{0,\,0,\,0,\,\frac{1}{2}\,\text{Sin}[\text{th}]^2\,\big(2+e^{-\text{la}[r,t]}\,\big(-2+r\,\text{la}^{(1,0)}[r,t]-r\,\text{nu}^{(1,0)}[r,t]\big)\big)\Big\}\Big\}$$

Die freien Einstein-Hilbert-Gleichungen verlangen, daß dieser Tensor verschwindet. Für die (01)- bzw. (10)-Komponente folgt daraus

```
la[r_, t_] = la[r];
```

Lassen wir uns die daraus folgenden Vereinfachungen für den Riccitensor anzeigen:

```
FullSimplify[RicciTen]
```

$$\left\{\left\{\frac{1}{4\,r}c^2\,e^{-la[r]+nu[r,t]}\left(nu^{(1,0)}[r,t]\left(4-r\,la'[r]+r\,nu^{(1,0)}[r,t]\right)+2\,r\,nu^{(2,0)}[r,t]\right),0,0,0\right\},\left\{0,\frac{1}{4\,r}\left(la'[r]\left(4+r\,nu^{(1,0)}[r,t]\right)-r\left(nu^{(1,0)}[r,t]^2+2\,nu^{(2,0)}[r,t]\right)\right),0,0\right\},\left\{0,0,1+\frac{1}{2}e^{-la[r]}\left(-2+r\,la'[r]-r\,nu^{(1,0)}[r,t]\right),0\right\},\left\{0,0,0,\frac{1}{2}\mathrm{Sin}[th]^2\left(2+e^{-la[r]}\left(-2+r\,la'[r]-r\,nu^{(1,0)}[r,t]\right)\right)\right\}\right\}$$

Durch Kombination der Gleichungen für die (00)- und die (11)-Komponente folgt, daß nu+lambda=f(t) ist. Diese Funktion f(t) können wir durch die noch freigestellte Umparametrisierung der Zeit t=t(t') zum Verschwinden bringen. Es ist also

```
nu[r_, t_] = nu[r]; la[r_] = -nu[r];
```

Der Ricci-Tensor vereinfacht sich dadurch gleich nochmals drastisch:

```
FullSimplify[RicciTen]
```

$$\left\{\left\{\frac{1}{2\,r}c^2\,e^{2\,nu[r]}\left(nu'[r]\left(2+r\,nu'[r]\right)+r\,nu''[r]\right),0,0,0\right\},\left\{0,-\frac{nu'[r]\left(2+r\,nu'[r]\right)+r\,nu''[r]}{2\,r},0,0\right\},\left\{0,0,1-e^{nu[r]}\left(1+r\,nu'[r]\right),0\right\},\left\{0,0,0,-\mathrm{Sin}[th]^2\left(-1+e^{nu[r]}\left(1+r\,nu'[r]\right)\right)\right\}\right\}$$

Lösen wir weiter zuerst die Gleichung für die (22)-Komponente (dann ist offenbar auch die Gleichung für die (11)-Komponente automatisch erfüllt:

```
DSolve[RicciTen[[3]][[3]] == 0, nu[r], r]
```

Solve::ifun: Inverse functions are being used by Solve, so
some solutions may not be found; use Reduce for complete solution information. ≫

$$\left\{\left\{nu[r]\to \mathrm{Log}\left[1-\frac{e^{C[1]}}{r}\right]\right\}\right\}$$

In der üblichen Notation mit dem Schwarzschildradius als Parameter ist

```
nu[r_] = Log[1 - rs / r]
```

$$\mathrm{Log}\left[1-\frac{rs}{r}\right]$$

Jetzt müssen wir noch nachprüfen, ob damit auch die verbleibenden Componenten des Ricci-Tensors verschwinden, so daß wir wirklich eine Lösung der Einstein-Hilbertgleichungen für das Vakuum gefunden haben

```
FullSimplify[RicciTen]
```

{{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}

Lassen wir uns also nochmal die endgültige Metrik in der Form des invarianten Viererlängenelements ausgeben

```
dq.g.dq
```

$$-dth^2\,r^2-\frac{dr^2}{1-\frac{rs}{r}}+c^2\,dt^2\left(1-\frac{rs}{r}\right)-dph^2\,r^2\,\mathrm{Sin}[th]^2$$

Das war historisch die erste exakte Lösung der freien Einstein-Hilbert-Gleichungen für das Gravitationsfeld im Rahmen der ART und wurde 1916 von Karl Schwarzschild gefunden. Daß die vermeintliche Singularität bei r=rs nur eine Koordinatensingularität ist, sieht man unter anderem daran, daß die Determinante

Det[g]

$-c^2 r^4 \mathrm{Sin}[th]^2$

bei r=rs nicht singulär wird. Die Schwarzschildlösung besitzt nur bei r=0 eine echte Raumzeit-Singularität. Das kann man durch die Koordinatentransformation zu Kruskal-Koordinaten zeigen (vgl. z.B. das Lehrbuch von E. Rebhan, Bd. 6). Auch die Singularität der Metrik entlang der “z-Achse” (entsprechend th=0 bzw. th=Pi) ist nur die übliche Koordinateningularität von Kugelkoordinaten.

Für gewöhnliche Himmelskörper tritt i.a. kein Problem beim Schwarzschildradius auf, weil dieser im Inneren der Körper gilt, wo man die Schwarzschildlösung für die Einstein-Hilbert-Gleichungen mit Materie aufsuchen muß. Ein Beispiel ist die Oppenheimer-Tolman-Volkoff-Lösung für statische Sterne. Dabei ergibt sich als Stabilitätskriterium, daß der Schwarzschildradius im Sterninneren liegen muß (s. wieder Rebhan, Bd. 6). Objekte, für die dies nicht erfüllt ist, sind ein schwarzes Loch, also eine punktförmige Singularität der Raumzeit, die nach dem sog. “no-hair theorem” von Wheeler im Rahmen der reinen ART-Maxwell-Theorie keinerlei Charakteristica als deren fundamentale Paramater (Masse, elektrische Ladung, Drehimpuls) aufweist.